


```
In[1]:= Trefoil = X[a, d, b, e] X[e, b, f, c] X[c, f, d, a]
```

```
Out[1]= X[a, d, b, e] X[c, f, d, a] X[e, b, f, c]
```

The definition of the Kauffman Bracket is given by the skein relations



and

$$\langle P \bigcirc \rangle \Rightarrow d \langle P \rangle$$

The first rule can be coded as follows :

```
In[2]:= rule1 = (X[a_, b_, c_, d_] :=> A δ[a d] δ[b c] + B δ[a b] δ[c d]);
```

Let us apply the first rule to the trefoil knot, just as an example:


```
In[3]:= Trefoil /. rule1
```

```
Out[3]= (A δ[b d] δ[a e] + B δ[a d] δ[b e])
         (A δ[c e] δ[b f] + B δ[b e] δ[c f]) (B δ[a d] δ[c f] + A δ[a c] δ[d f])
```

```
In[4]:= Trefoil /. rule1 // Expand
```

```
Out[4]= A^2 B δ[a d] δ[b d] δ[a e] δ[c e] δ[b f] δ[c f] + A B^2 δ[a d]^2 δ[b e] δ[c e] δ[b f] δ[c f] +
         A B^2 δ[a d] δ[b d] δ[a e] δ[b e] δ[c f]^2 + B^3 δ[a d]^2 δ[b e]^2 δ[c f]^2 +
         A^3 δ[a c] δ[b d] δ[a e] δ[c e] δ[b f] δ[d f] + A^2 B δ[a c] δ[a d] δ[b e] δ[c e] δ[b f] δ[d f] +
         A^2 B δ[a c] δ[b d] δ[a e] δ[b e] δ[c f] δ[d f] + A B^2 δ[a c] δ[a d] δ[b e]^2 δ[c f] δ[d f]
```

We see that we need a rule that will dispose of unnecessary intermediate points :



```
In[5]:= rule2 = (δ[a_b_] δ[b_c_] :=> δ[a c]);
```

```
In[6]:= (Trefoil /. rule1 // Expand) //. rule2
```

```
Out[6]= A^2 B δ[c f]^2 + A B^2 δ[a d]^2 δ[c f]^2 + A B^2 δ[b e]^2 δ[c f]^2 +
        B^3 δ[a d]^2 δ[b e]^2 δ[c f]^2 + 2 A^2 B δ[d f]^2 + A B^2 δ[b e]^2 δ[d f]^2 + A^3 δ[c e]^2 δ[d f]^2
```

This done, we can code the second rule, (using dd instead of d, to avoid naming conflicts)

```
In[7]:= rule3 = {(δ[_])^2 -> dd, δ[_^2] -> dd};
```


and see what happens to the trefoil knot:

```
In[8]:= RawBracket[t_] := Simplify[(t /. rule1 // Expand) //. rule2 /. rule3]
```

```
In[9]:= RawBracket[Trefoil]
```

```
Out[9]= dd (3 A^2 B + A^3 dd + 3 A B^2 dd + B^3 dd^2)
```

Let's verify the second Reidemeister move :



```
In[10]:= RawBracket[X[b, e, f, a] X[f, e, c, d]] == δ[a d] δ[b c]
```


```
Out[10]= A B δ[b c] δ[a d] + (A^2 + B^2 + A B dd) δ[a b] δ[c d] == δ[b c] δ[a d]
```

Thus we see that for the second Reidemeister move to hold, we must have  $AB = 1$  and  $A^2 + B^2 + ABd = 0$ . Solving for  $B$  and  $dd$  in terms of  $A$  we get:

```
In[11]:= rule4 = {B -> 1/A, dd -> -A^2 - 1/A^2}
```

```
Out[11]= {B -> 1/A, dd -> -1/A^2 - A^2}
```

Time to check the third Reidemeister move: (It's guaranteed to hold, but loose nothing by checking again)



```
In[12]:= Bracket[t_] := Simplify[RawBracket[t] / dd /. rule4]
```


```
In[13]:= Bracket[X[a, b, g, i] X[g, c, d, h] X[i, h, e, f]]
```

```
Out[13]= -\frac{1}{1+A^4} (A (\delta[a b] \delta[d e] \delta[c f] + \delta[c d] (A^4 \delta[b e] \delta[a f] + A^2 \delta[a b] \delta[e f]) + \delta[b c] (A^2 \delta[d e] \delta[a f] + \delta[a d] \delta[e f])))
```

```
In[14]:= Bracket[X[b, c, i, h] X[d, e, g, i] X[g, f, a, h]]
```

```
Out[14]= -\frac{1}{1+A^4} (A (\delta[a b] \delta[d e] \delta[c f] + \delta[c d] (A^4 \delta[b e] \delta[a f] + A^2 \delta[a b] \delta[e f]) + \delta[b c] (A^2 \delta[d e] \delta[a f] + \delta[a d] \delta[e f])))
```

The remaining sad point is the behavior of the bracket under the first Reidemeister move :



```
In[15]:= Simplify[RawBracket[X[c, a, b, c]] /. rule4]
```

```
Out[15]= -A^3 \delta[a b]
```

This is unfortunate, but not fatal. It can be fixed by multiplying the bracket by  $-A^3$  raised to an appropriate multiple of the writhe is common to make the further substitution  $A \rightarrow q^{1/4}$  and then the resulting invariant is called the Jones polynomial. Let us

```
In[16]:= Jones[t_, w_] := Simplify[(Bracket[t] (-A^3)^(-w)) /. A -> q^(1/4)]
```

```
In[17]:= Jones[Trefoil, 3]
```

```
Out[17]= \frac{-1 + q + q^3}{q^4}
```

```
In[18]:= MirrorTrefoil = Trefoil /. x_X -> RotateLeft[x]
```

```
Out[18]= X[b, f, c, e] X[d, b, e, a] X[f, d, a, c]
```

```
In[19]:= Jones[MirrorTrefoil, -3]
```

```
Out[19]= q + q^3 - q^4
```

Ok, if you insist,

```
In[20]:= w[K_] := Module[
  {p, l},
  l = ReplacePart[K /. X[a_, b_, c_, d_] -> \delta[a c] \delta[b d], p, {1, 0}] //.
    {p[a_b_] -> p[a, b], p[a_, b_, c_, d_] -> p[a, b, c, d]};
  l = MapThread[Rule, {List @@ l, Range[Length[l]]}];
  (Plus @@ (K /. l)) /.
  X[a_, b_, c_, d_] -> If[Abs[p = (a - c) (b - d)] == 1, p, -Sign[p]]
]
```

```
In[21]:= Jones[K_] := Jones[K, w[K]]
```

```
In[22]:= {Jones[Trefoil], Jones[MirrorTrefoil]}
```

```
Out[22]= { $\frac{-1 + q + q^3}{q^4}$ ,  $q + q^3 - q^4$ }
```

Now, last but not least,

```
In[23]:= Expand[ (
    q * (-A^3)^(-1) * RawBracket[X[a, b, c, d]] -
    q^(-1) * (-A^3) * RawBracket[X[b, c, d, a]]
) /. {A -> q^(1/4), B -> q^(-1/4)}
]
```

```
Out[23]=  $\frac{\delta[b c] \delta[a d]}{\sqrt{q}} - \sqrt{q} \delta[b c] \delta[a d]$ 
```